
Abstract

I present and analyze a version of Quicksort for linked lists without “end”-pointers.
On random lists its execution speed compares favorably to that of an efficient list
version of Merge sort.

1 Specification

While working on a small graphics project, I got to a point where linked lists had to
be sorted (on depth coordinate for the painter’s algorithm [FvDFH90]). Merge sort is
very appropriate for sorting lists. Nevertheless, I wondered whether I could also employ
Quicksort, since for arrays Quicksort is on the average about 25% faster than Merge sort
[Meh84, p. 67]. Knuth [Knu73] considers several sorting algorithms for linked lists, but
Quicksort is not among them.

Before embarking on a specification of the problem, let me explain my notation
for (finite abstract) lists, adapted from [BW88]. The list consisting of then elements
a0, . . . ,an−1 (in this order) is denoted by [a0, . . . ,an−1]. In particular, [ ] denotes the

empty list and [a] denotes the list consisting of elementa only. The catenation of listss
andt (in this order) is denoted bys++ t . For elementa and lists, I abbreviate [a] ++ s
to a : s. An infix dot stands for function application; it binds stronger than catenation.
Function # returns the length of its list argument, for example #.(s++ t) = #.s+ #.t .
Finally, sort is a function from lists to lists such thatsort.s consists of the elements ofs
in ascending order.

type ElementType= . . . ;
List = ↑Cell ;
Cell= record

value: ElementType;
next: List
end { Cell } ;

Figure 1: Type definitions to implement lists

The Pascal type definitions in Figure 1 form the basis of an implementation of lists
over some element type. For convenience’ sake, I assume thatElementTypeis ordered
by<. For listss andt overElementType, I write s< t to express that each elements ofs
precedes all elements oft .

The typeList represents the set of lists overElementType. In order to express the
representation invariant and abstraction function, it is convenient to define some auxiliary
operators onList (these are adapted from Lex Bijlsma’s course notes for Programming 3).
for p andq of typeList and integern (n ≥ 0), define

drop.p.n =
{

p if p = nil ∨ n = 0
drop.(p↑.next).(n− 1) if p 6= nil ∧ n > 0

p→ q = (∃n : n ≥ 0 : drop.p.n = q)

p F q =
{

[ ] if p = q
(p↑.value) : (p↑.nextF q) if p 6= q

Mappingdrop walks along thenext-field, dropping an initial segment of the chain. Pred-
icate p→ q expresses thatq is reachable fromp. If p→ q holds thenp F q is the list
of values in the chain fromp to q.

The representation invariantI and abstraction function [[ ]] on the typeList are now
defined by

I .s = s→ nil

[[s]] = s F nil

Note the following properties of the abstraction function.

[[ p]] = [ ] ≡ p = nil

[[ p]] = a : s ≡ p↑.value= a ∧ [[ p↑.next]] = s

The specification of the sorting problem can be expressed by

function Quicksort(s: List): List ;
{ assumes [[s]] = S ; returnsz, where [[z]] = sort.S }

2 Design

The “standard” Pascal design forQuicksortis shown in Figure 2. The problem with this
standard approach is that catenation of arbitraryList values cannot be done in constant
time. Note that another representation of lists, viz. with also a pointer to the last element,
would solve this problem. However, in my project the typeList was given and could not
be tinkered with. Anyway, the use of “end”-pointers would not have been an improve-
ment, since either they have to be determined on the fly by walking through the list, or
they have to be maintained by all other operations as well.

function Quicksort(s: List): List ;
{ assumes [[s]] = S ; returnsz, where [[z]] = sort.S }
var l , r : List ;
begin
if s= nil then { [[s]] = [ ] }

Quicksort:= s
else withs↑ do begin{ [[s]] = value: [[next]] }

“partition [[next]], usingvalueas pivot” ;
{ sort.S= sort.([[ l ]] ++ [value] ++ [[r ]]) ∧ [[ l ]] < [value] ≤ [[r ]] }
{ hence,sort.S= sort.[[ l ]] ++ [value] ++ sort.[[r ]] }
next := Quicksort(r ) ; { sort.S= sort.[[ l ]] ++ [[s]] }
Quicksort:= “catenation ofQuicksort(l ) ands”
end { else withs↑ }

end { Quicksort} ;

Figure 2: “Standard” design for Quicksort in Pascal

In a functional style [BW88, p. 154],Quicksortcan be defined by

Quicksort.[ ] = [ ]

Quicksort.(a : u) = Quicksort.l ++ [a] ++Quicksort.r
where l = [x | x← u; x < a]

r = [x | x← u;a ≤ x]

Here, [x | x ← u; x < a] is the list of elementsx drawn fromu that are less thana
(see [BW88, p. 50]). In functional terms, my goal is to eliminate++ in favor of the more
efficient ‘:’ operator.

The solution is based on the observation thatQuicksortsatisfies

Quicksort.s = Quicksort.s++ [ ]

and, therefore, its specification can be generalized toQsortby defining

Qsort.s.t = Quicksort.s++ t

Specialization of this definition yieldsQuicksort.s= Qsort.s.[ ]. For Qsort, I derive

Qsort.[ ] .t = Quicksort.[ ] ++ t
= [ ] ++ t
= t

and, assumingl = [x | x← u; x < a] andr = [x | x← u;a ≤ x],

Qsort.(a : u).t = Quicksort.(a : u)++ t

= Quicksort.l ++ [a] ++Quicksort.r ++ t

= Quicksort.l ++ (a : Quicksort.r ++ t)

= Qsort.l .(a : Qsort.r.t)

Thus, a functional definition forQsort is

Qsort.[ ] .t = t

Qsort.(a : u).t = Qsort.l .(a : Qsort.r.t)
where l = [x | x← u; x < a]

r = [x | x← u;a ≤ x]

Figure 3 gives the corresponding Pascal implementation ofQsort. I have coded the parti-
tion phase explicitly.

The Pascal implementation ofQuicksortnow boils down to

function Quicksort(s: List): List ;
{ assumes [[s]] = S ; returnsz, where [[z]] = sort.S }
begin Quicksort:= Qsort(s,nil) end ;

Of course, this list version of Quicksort still has quadratic worst-case time-complexity.
Furthermore, it is not stable (in the sense that records with equal values retain their origi-
nal relative ordering [Knu73, p. 4]), because each partitioning reverses the order (the lists
are manipulated as stacks). Quicksort for arrays is, in general, also not stable. Stabil-
ity can be guaranteed, but at the price of complicating the partitioning phase (also see
[Mot81, Weg82], where “end”-pointers are used). In Section 4, I suggest another version
of Quicksort for linked lists. That version is stable and also allows limitation of the max-
imum recursion depth. The latter cannot easily be accomplished in the Pascal version
of Qsort, unless the compiler implements tail recursion efficiently or allows pointers to
static variables. A final remark aboutQsort is that the head of the list is used as partition-
ing pivot and that taking another element (for instance, a random element or the median
of three random elements as is often done for improvement) would be costly, since lists
cannot be indexed efficiently.

function Qsort(s, t : List): List ;
{ assumes [[s]] = S ∧ [[ t ]] = T ; returnsz, where [[z]] = sort.S++ T }
var a: ElementType; u, l , r,h: List ;
begin
if s= nil then Qsort := t
else withs↑ do begin

a := value; u := next; { [[s]] = a : [[u]] }
l := nil ; r := nil ;
{ invariant:sort.S= sort.([[ l ]] ++ [a] ++ [[r ]] ++ [[u]]) ∧ [[ l ]] < [a] ≤ [[r ]] }
while u 6= nil do with u↑ do { [[u]] = value: [[next]] }

if value< a then { “ l ,u := value: l ,next” }
begin h := next; next := l ; l := u ; u := h end

else{ “r,u := value: r,next” }
begin h := next; next := r ; r := u ; u := h end ;

{ sort.S= sort.[[ l ]] ++ [a] ++ sort.[[r ]] }
next := Qsort(r, t) ; { sort.S= sort.[[ l ]] ++ [[s]] }
Qsort := Qsort(l , s)
end { else withs↑ }

end { Qsort} ;

Figure 3: Pascal implementation ofQsort

3 Efficiency

It is not guaranteed that the efficiency of the above list version of Quicksort is compa-
rable to that of the array version. One has to be very careful to extrapolate performance
figures [Sed78]. Therefore, it is interesting to compareQsort to a (good) list version of
Merge sort. In case of the array version, Quicksort is on the average 25% faster than
Merge sort [Meh84, p. 67]. For the list versions their execution speed on random lists
turns out to be comparable.

I consider the Pascal function for Merge sort given in Figure 4. It first splits the list
into two halves by cutting the list roughly in the middle (in thewhile-loop, u is roughly
halfway betweens and t). A common alternative is to unzip the list into even and odd
elements to obtain the two halves, but this turns out to be slightly more expensive and
stability is lost. The two halves are then sorted recursively and finally merged to yield the
result. The Boolean operatorscor andcandstand for conditionalor andand respectively
(also known as short-circuit boolean operators).

The functionMergeis defined in Figure 5. The precondition ofMergeexpresses that
its arguments should be non-intertwined non-empty sorted lists. This is the case for the
invocation inMergesortabove (however, invokingMerge(s, s) would get into one kind
of trouble or another). Of course, the invocation ofMergecan be unfolded into a loop to
obtain a more efficient program (with respect to both speed and memory usage). I leave
this to the reader. By the way, the above combination ofMergesortandMerge forms a

function Mergesort(s: List): List ;
{ assumes [[s]] = S ; returnsz, where [[z]] = sort.S }
var t,u: List ;
begin
if (s= nil) cor (s↑.next= nil) then Mergesort:= s
else begin{ #.S≥ 2 }

u := s ; t := u↑.next↑.next;
{ invariant: #.(s F u↑.next) = #.(u↑.nextF t) }
while (t 6= nil) cand (t↑.next 6= nil) do

begin u := u↑.next; t := t↑.next↑.nextend ;
t := u↑.next; u↑.next := nil ;
{ S= [[s]] ++ [[ t ]] ∧ 0≤ #.[[ t ]] − #.[[s]] ≤ 1, hence [[s]] 6= [ ] ∧ [[ t ]] 6= [ ] }
Mergesort:= Merge(Mergesort(s),Mergesort(t))
end { else}

end { Mergesort} ;

Figure 4: Pascal implementation of Merge sort

stable sorting algorithm.
I have compared the execution times on random lists for Quicksort (based onQsort,

with singleton lists done directly instead of by partitioning), Merge sort (based onMergesort,
with test s = nil eliminated (because the random lists are non-empty), and unfolded
Merge), and a list version of insertion sort. The results are presented in Table 1. The ex-
ecution times are relative to those for applying the identity function. The functions were
applied to the same data, by resetting the seed of the random generator before each test
run.

function Merge(s, t : List): List ;
{ assumes¬(∃q : q 6= nil : s→ q ∧ t → q) ∧ [[s]] = S ∧ [[ t ]] = T ∧ }
{ S 6= [ ] ∧ T 6= [ ] ∧ sort.S= S ∧ sort.T = T }
{ returnsz, where [[z]] = sort.(S++ T) }
begin
if s↑.value≤ t↑.valuethen with s↑ do { “Merge:= value: Merge(next, t)” }

if next= nil then begin next := t ; Merge:= s end
else beginnext := Merge(next, t) ; Merge:= s end

else with t↑ do begin{ “Merge:= value: Merge(s,next)” }
if next= nil then begin next := s ; Merge:= t end
else beginnext := Merge(s,next) ; Merge:= t end

end { Merge} ;

Figure 5: Pascal implementation ofMerge

List length 10 100 1000 10,000 100,000
Repeat count 10,000 1000 100 100 1
Quicksort 33 51 70 91 114
Merge sort 42 63 82 102 121
Insertion sort 11 86 846 8459 not run

Table 1: Comparison of execution times (seconds)

The advantage ofQuicksortis that it contains just one loop (for partitioning), whereas
Mergesortcontains two loops (one for splitting and one for merging). The disadvantage
of Quicksortis that partitioning may yield two lists whose lengths differ in order of mag-
nitude, whereasMergesortalways works with a perfectly ‘balanced’ split. For random
lists, the disadvantage is apparently less important than the advantage of a single loop.
The break-even point forQuicksortand insertion sort occurs at lists of about fifty el-
ements. For the array versions, the break-even point usually occurs at a much shorter
length [Sed78], because array insertion involves data movement in contrast to list inser-
tion.

4 Dummy Head Cells

The implementation of lists by the typeList has a small disadvantage, viz. that all ele-
mentsexcept possibly the first(if present) are pointed to from anext-field. This may give
rise to a case distinction in programs dealing with such lists (this happens, for example,
when maintaining stability inQsort or unfolding Merge into a loop). The exceptional
case can be avoided by prefixing each list with a dummy head cell. This is expressed in
the following type definition, and corresponding representation invariant and abstraction
function.

HeadedList = List

I .s = s 6= nil ∧ s→ nil

[[s]] = s↑.nextF nil

Each element in a list of typeHeadedListis pointed to from anext-field. In particular, the
first element is pointed to by thenext-field of the dummy head cell.

The version ofQsort in Figure 6 is designed in terms ofHeadedList. Partitioning
does not reverse the order and, hence,QsortH is stable. Its maximum recursion depth
has been limited by recursing only on the shorter of the two lists obtained by partitioning.
For that purpose, countersm andn have been introduced. Observe that the (dummy) head
cell of s is used as head cell of the “left” partitioning result, and the cell containing the
pivot is used as head cell of the “right” partitioning result. Thus, no additional cells are
needed.

Table 2 presents some timing results on random headed lists, again relative to apply-
ing the identity transformation. Entry ‘Ltd. Quicksort’ is based onQsortHwith singleton
lists done directly. Entry ‘Quicksort’ is like ‘Ltd. Quicksort’ but without limiting the

procedure QsortH(s, t : HeadedList) ;
{ assumes [[s]] = S ∧ [[ t ]] = T ∧ s↑.value= A }
{ establishes [[s]] = sort.S++ T ∧ s↑.value= A }
var a: ElementType; r : HeadedList; sl, rl ,u: List ; m,n: Integer;
begin
while s↑.next 6= nil do begin

r := s↑.next; u := r↑.next; a := r↑.value; { [[s]] = a : [[u]] }
sl := s ; rl := r ; { abbreviateL = s↑.nextF sl↑.nextandR= r↑.nextF rl ↑.next}
m := 0 ; n := 0 ; { m= #.L andn = #.R }
{ invariant:sort.S= sort.(L ++ [a] ++ R++ [[u]]) ∧ L < [a] ≤ R }
while u 6= nil do with u↑ do { [[u]] = value: [[next]] }

if value< a then { “ L ,u := L ++ [value],next” }
begin sl↑.next := u ; sl := u ; m := m+ 1 ; u := nextend

else{ “ R,u := R++ [value],next” }
begin rl↑.next:= u ; rl := u ; n := n+ 1 ; u := nextend ;

sl↑.next:= nil ; rl↑.next := nil ;
{ sort.S= sort.[[s]] ++ [a] ++ sort.[[r ]] }
if m< n then beginQsortH(s, r ) ; s := r end
else beginQsortH(r, t) ; t := r end
end { while } ;

s↑.next := t
end { Qsort} ;

Figure 6: Version ofQsort in terms ofHeadedList

maximum recursion depth. Entry ‘Merge sort’ is adapted from Merge sort in the preced-
ing section. Note that each iteration through the partitioning loop ofQSortHconsists of

List length 10 100 1000 10,000 100,000
Repeat count 10,000 1000 100 100 1
Ltd. Quicksort 29 47 68 90 114
Quicksort 32 49 67 85 105
Merge sort 44 66 87 107 127

Table 2: Comparison of execution times (seconds) for headed lists

two comparisons and four assignments, the same as inQSort. This explains why ‘Ltd.
Quicksort’ for headed lists performs almost the same as Quicksort in Table 1, and also
why Quicksort in Table 2 performs better (it has one assignment fewer in its partition-
ing loop). Merge sort in Table 2 performs less well that in Table 1, because only the
(dummy) head cell of its argument is available (no pivot), and thus the second half (of the
split) must be attached to that same head cell before recursing on it.

5 Concluding Remarks

I have presented a list version of Quicksort for linked lists, calledQsort. The lists have
no “end”-pointer to the last element. The Pascal programQsort is very short. On random
lists it performs well compared to a list version of Merge sort. There are however a few
side remarks to be made.

First of all, Qsort is not stable. On the other hand, most versions of Quicksort and
some versions of Merge sort are also not stable. Secondly, it is harder to implement a
version that does not use the head of the list as pivot (but, for instance, a random element).
In Pascal it is also harder to limit the maximum recursion depth by putting the “longest”
recursive call at the end and transforming it into a loop. However, when using lists with
a dummy head cell, it is easy to guarantee stability and to limit the maximum recursion
depth.

In the same vein, a list version of Quicksort is feasible that operates in constant-space
(i.e. without recursion overhead) [Ver93].

I would like to thank Johan Lukkien and the Eindhoven Algorithm Club led by Anne
Kaldewaij for reading and commenting on an earlier version of this note.

